

โปรแกรมย่อย (Subprogram)

รศ.ณรงค์ บวบทอง

Outline

- Subprograms
- Functions
- Argument Classes
- Procedures
- Overloading
- Visibility of Declarations

โปรแกรมย่อย (Subprogram)

โปรแกรมย่อยคือกลุ่มของคำสั่งลำดับที่สามารถเรียกใช้งานได้ โปรแกรมย่อยนี้สามารถเขียนไว้ในส่วนของ Package หรือ Architecture หรือ Process ก็ได้ แต่การเรียกใช้งานจะขึ้นอยู่กับตำแหน่งที่ประกาศโปรแกรมย่อยไว้ดังนี้

- ประกาศไว้ใน Package สามารถเรียกใช้ได้จากทุกๆที่ของโมเดล (Model)
- ประกาศไว้ใน Architecture ของโมเด็ลใดก็สามารถเรียกใช้ได้จากภายในโมเด็ลนั้นเท่านั้น
- ประกาศไว้ใน Process ใดก็สามารถเรียกใช้ได้จากภายใน Process นั้นเท่านั้น

โปรแกรมย่อย

ภาษา VHDL แบ่งโปรแกรมย่อยออกเป็น 2 ประเภทคือ

- ฟังก์ชั่น (Function)
- โปรซีเยอร์ (Procedure)

ฟังก์ชั่น (Function)

- ฟังก์ชั่นเป็นโปรแกรมย่อยชนิดหนึ่ง เมื่อเรียกใช้จะส่งค่ากลับคืนเพียงหนึ่งค่า ทั้งนี้การเรียกใช้อาจมีการส่งค่าพารามิเตอร์ (Passing parameter) ให้กับฟังก์ชั่นหรือไม่ก็ได้ แต่ถ้ามีการส่งค่าผ่านเข้าสู่ฟังก์ชั่น พารามิเตอร์นั้น ต้องไม่มีการเปลี่ยนแปลงค่า แต่ค่าของพารามิเตอร์จะถูกนำไปใช้เพื่อคำนวณหาผลลัพธ์ โดยผลลัพธ์นี้จะถูกส่งกลับไปยังแบบจำลองที่เรียกใช้ตรงตำแหน่งที่ฟังก์ชั่นนั้นถูกเรียกใช้ ลักษณะของฟังก์ชั่นมีคุณสมบัติดังนี้

คุณสมบัติของฟังก์ชั่น

- การทำงานเป็น นิพจน์(expression) ไม่ใช่ statement
- ฟังก์ชั่นจะคำนวณให้ผลลัพธ์เพียงค่าเดียว และส่งกลับคืนแบบจำลอง
- ภายในฟังก์ชั่นต้องมีคำสั่ง RETURN เสมอ
- การทำงานของฟังก์ชั่นต้องไม่เปลี่ยนแปลงค่าออบเจ็กต์ (Object) ที่ถูกส่งผ่าน
- ฟังก์ชั่นต้องประกอบด้วย **function body** และอาจจะมี **function declaration** ด้วยก็ได้

Function declaration

- เป็นส่วนที่อาจจะมีหรือไม่ก็ได้ มีรูปแบบดังนี้

```
Function name [(Formal_parameter_list)] Return type;
```

name เป็นชื่อของฟังก์ชั่น

formal parameter list เป็นรายชื่อของพารามิเตอร์ที่ฟังก์ชั่นต้องการให้ส่งผ่าน สามารถกำหนดคุณสมบัติเพิ่มเติมให้พารามิเตอร์ได้ เช่น CLASS, MODE และ TYPE

RETURN ใช้กำหนด TYPE ของค่าที่จะส่งกลับ

เช่น FUNCTION bl2bit (a : boolean) RETURN BIT;

Function body

- เป็นส่วนที่บรรยายพฤติกรรมของฟังก์ชั่น ถ้าจะให้ฟังก์ชั่นทำอะไรส่วนนี้จะต้องมี ภายใน body เป็น คำสั่งลำดับ ห้ามมีคำสั่ง concurrent และต้องมีคำสั่ง RETURN ซึ่งเป็นคำสั่งให้ส่งค่ากลับและเป็นคำสั่งสิ้นสุดการทำงานด้วย รูปแบบของ function body เป็นดังนี้

```
FUNCTION name [(formal_parameter_list)] RETURN TYPE IS
        declaration statement
BEGIN
        [sequential statements]
        [include RETURN expression]
END name;
```

ตัวอย่าง

```
FUNCTION bl2bit (a : BOOLEAN) RETURN BIT IS
BEGIN
      IF a then
         RETURN '1';
      ELSE
         RETURN '0';
      END IF;
END bl2bit;
```

ตัวอย่างฟังก์ชั่น bl2bit

```
-- i2bv : Integer to Bit_vector.
-- In :  Integer, Value and width.
-- Return : Bit_vector, with left bit is the most significant bit.
package my_package is
            FUNCTION bl2bit (a : boolean) RETURN BIT;
end my_package;
package body my_package is
    FUNCTION bl2bit (a : BOOLEAN) RETURN BIT IS
    BEGIN
            IF a then
                        RETURN '1';
            ELSE
                        RETURN '0';
            END IF;
    END bl2bit;
end my_package;
```

Function Declaration

Function Body

ตัวอย่างฟังก์ชั่น bl2bit แบบที่ 2

```
-- i2bv : Integer to Bit_vector.
-- In :  Integer, Value and width.
-- Return : Bit_vector, with left bit is the most significant bit.
package my_package is
        function i2bv (val, width : integer) return bit_vector;
end my_package;
package body my_package is
        function i2bv (val, width : integer) return bit_vector is
                variable result : bit_vector( width-1 downto 0) := (others => '0');
        variable bits : integer := width;
        begin
                for i  in 0 to bits-1 loop
                        if ( (val/(2**i) ) mod 2 = 1) then
                                result(i) := '1';
                        end if;
                end loop;
                return (result);
        end i2bv;
end my_package;
```

Function Declaration

Function Body

การเรียกใช้ฟังก์ชั่น bl2bit ใน package สามารถทำได้ดังนี้

```
-- cnt4 : 4-bit binary counter .
-- model : behavioral
use work.my_package.all;
entity cnt4 is
        port(dout : out bit_vector(3 downto 0);
        clk, reset  : in bit);
end cnt4;
architecture beh of cnt4 is
begin
        counter: process(clk, reset)
        variable cnum : integer;
        begin
                if reset = '1' then
                        cnum := 0;
                elsif (clk = '1') then
                        cnum := cnum + 1;
                end if;
                dout <= i2bv(cnum,4);
        end process;
end beh;
```

ตัวอย่างฟังก์ชั่นสำหรับวงจร Full adder เมื่อเขียนอยู่ใน Package

```
library ieee;
use ieee.std_logic_1164.all;
package my_package is
        function FULLADD (A,B, CIN : std_logic )  return std_logic_vector;
end my_package ;
package body my_package is
        function FULLADD (A,B, CIN : std_logic )  return std_logic_vector is
                variable SUM, COUT : std_logic;
                variable RESULT : std_logic_vector  (1 downto 0);
        begin
                SUM := A xor B xor CIN;
                COUT := (A and B) or (A and CIN) or (B and CIN);
                RESULT := COUT & SUM;
                return RESULT;
        end FULLADD;
end my_package ;
```

การเรียกใช้ฟังก์ชั่น FULLADD ใน package

```
library ieee;
use ieee.std_logic_1164.all;
use work. my_package.all;
entity EXAMPLE is
        port ( A,B : in std_logic_vector (3 downto 0);
                CIN : in std_logic;
                S : out std_logic_vector (3 downto 0);
                COUT : out std_logic );
end EXAMPLE;
architecture ARCHI of EXAMPLE is
        signal S0, S1, S2, S3 : std_logic_vector (1 downto 0);
        begin
                S0 <= FULLADD (A(0), B(0), CIN);
                S1 <= FULLADD (A(1), B(1), S0(1));
                S2 <= FULLADD (A(2), B(2), S1(1));
                S3 <= FULLADD (A(3), B(3), S2(1));
                S <= S3(0) & S2(0) & S1(0) & S0(0);
                COUT <= S3(1);
end ARCHI;
```

Recursive Function

- หมายถึงฟังก์ชั่นที่สามารถเรียกใช้ตัวเองได้ เช่นตัวอย่างฟังก์ชั่นการหาค่า factorial ของจำนวน n ใดๆ สามารถเขียนเป็นฟังก์ชั่นได้ดังนี้

```
package my_package is
        function my_factorial(x : integer) return integer;
end my_package ;
package body my_package is
        function my_factorial(x : integer) return integer is
        begin
                if x = 1 then
                        return x;
                else
                        return (x*my_factorial(x-1));
                end if;
        end function my_factorial;
end my_package ;
```

การเรียกใช้ฟังก์ชั่น my_factorial ใน package

```
use work. my_package.all;
entity EXAMPLE_3 is
        port (A : in integer range 0 to 255;
                Y : out integer range 0 to 255);
 end EXAMPLE_3;
architecture ARC of EXAMPLE_3 is
begin
        S <= my_factorial(A);
end ARC;
```

โปรซีเยอร์ (Procedure)

- ทั้ง ฟังก์ชั่น และโปรซีเยอร์ ต่างก็เป็นโปรแกรมย่อย (Subprogram) ที่มีวัตถุประสงค์ของการเขียนเหมือนกัน คือ ต้องการรวบรวมคำสั่งแบบลำดับที่มีการเรียกใช้บ่อยๆ เอาไว้ให้เรียกใช้ได้สะดวกโดยไม่ต้องเขียนคำสั่งนั้นทุกๆครั้งที่ต้องการใช้งาน โครงสร้างของโปรซีเยอร์ก็คล้ายกับของฟังก์ชั่น

ความแตกต่างระหว่างฟังก์ชั่นกับโปรซีเยอร์

- **ฟังก์ชั่น ไม่สามารถเปลี่ยนแปลงค่าที่ส่งเข้าไปได้ แต่โปรซีเยอร์ทำได้**
- **ฟังก์ชั่นมี RETURN เพื่อส่งค่าคืน แต่โปรซีเยอร์มีหรือไม่ก็ได้ ถ้ามีอาจเพียงเป็นการหยุดการทำงานของโปรซีเยอร์**
- **ฟังก์ชั่นคืนค่ากลับเพียงค่าเดียว แต่โปรซีเยอร์สามารถส่งคืนได้หลายค่า**
- **ทิศทางหรือ Mode ของพารามิเตอร์ที่ส่งผ่าน (formal parameter list) เข้าสู่ฟังก์ชั่นเป็นประเภท IN เท่านั้น แต่ของโปรซีเยอร์เป็นได้ทั้ง IN, OUT และ INOUT**
- **การเรียกฟังก์ชั่นเป็นการเรียกแบบ Expression แต่ในโปรซีเยอร์เป็นการเรียกแบบ Statement**
- **เช่นเดียวกับฟังก์ชั่น โปรซีเยอร์ประกอบด้วย Procedure declaration และ Procedure body**

Procedure Declaration

```
PROCEDURE  name (formal_parameter_list);
```

- **name** ชื่อของ โปรซีเยอร์
- **formal_parameter_list** รายชื่อพารามิเตอร์ที่ ใช้ใน โปรซีเยอร์ มีหน้าที่เป็นตัวบอก CLASS, NAME, MODE และ TYPE ของ objec และมีรูปแบบการเขียนเป็นดังนี้

```
(CLASS object_name : MODE TYPE)
```

(CLASS object_name : MODE TYPE)

- **CLASS** หมายถึงชั้นของ object ซึ่งอาจจะเป็น SIGNAL, VARIABLE หรือ CONSTANT ถ้าไม่มีการกำหนด CLASS และ MODE ภาษา VHDL จะถือว่าเป็น CONSTANT MODE IN แต่ถ้าไม่กำหนด CLASS แต่กำหนด MODE เป็น IN จะถือว่าเป็น VARIABLE
- **Object_name** ชื่อของ object
- **MODE** ทิศทางการไหลของข้อมูล มีสามชนิดคือ
  - **IN** ค่าของพารามิเตอร์ที่ส่งเข้าโปรซีเยอร์ แก้ไขหรือเปลี่ยนแปลงไม่ได้
  - **OUT** ค่าของพารามิเตอร์ที่ส่งออกจากโปรซีเยอร์
  - **INOUT** ค่าของพารามิเตอร์ที่ส่งเข้าได้รับคืนจากโปรซีเยอร์ แก้ไขหรือเปลี่ยนแปลงได้
- **TYPE** เป็นตัวกำหนดกลุ่มของค่าต่างๆที่ object สามารถมีได้

ตัวอย่างของโปรซีเยอร์ declaration ที่อยู่ใน Package declaration



Procedure Body

- ในส่วนของ body นี้ ประกอบด้วยลำดับของ Sequential statement ที่บรรยายความสัมพันธ์ระหว่างค่าของ input parameter กับค่าที่จะส่งกลับ ในส่วนนี้ห้ามมี Concurrent statement ในโปรซีเยอร์ไม่จำเป็นต้องมีคำสั่ง RETURN คำสั่ง RETURN มีหน้าที่หยุดการทำงานของโปรซีเยอร์ สำหรับโครงสร้างที่ไม่มีคำสั่ง RETURN การทำงานของโปรซีเยอร์จะหยุดได้ด้วยคำสั่ง END ในบรรทัดสุดท้ายของโปรซีเยอร์

รูปแบบโครงสร้าง procedure body

```
PROCEDURE name (formal_parameter_list) IS
        -- declarative_statments
BEGIN
        --sequential_statements
END name;
```

ตัวอย่าง Procedure body

```
PACKAGE BODY util IS
        PROCEDURE add_element(element : IN REAL;
                        VARIABLE filter_data : INOUT filter_data_type ) IS
        BEGIN
                FOR IN filter_data'HIGH DOWNTO filter_data' LOW+1 LOOP
                        filter_data(i) := filter_data (i-1);
                END LOOP;
                filter_data(filter_data'LOW) := element;
        END add_element;

        PROCEDURE zero_out(input : INOUT filter_data_type) IS
        BEGIN
                FOR i IN input'RANGE LOOP
                        input(i) := 0.0;
                END LOOP;
        END zero_out;
        PROCEDURE still_busy IS
        BEGIN
                ASSERT FALSE REPORT "Still Busy!" SEVERITY NOTE;
        END stil_busy;
END util;
```

Procedure Calls

- การเรียกโปรซีเยอร์มาใช้ ทำได้โดยการเขียนชื่อ (name) ของ โปรซีเยอร์ นั้น

```
procedure_name (passing_parameter_list);
```

passing_parameter_list

- สัมพันธ์โดยตำแหน่ง (Position Association) แบบนี้ให้เขียนชื่อพารามิเตอร์ที่ต้องการส่งผ่านให้ตรงกับตำแหน่งพารามิเตอร์ที่เป็นตัวรับ เช่นถ้าโปรซีเยอร์ชื่อ FULLADD

```
PROCEDURE  FULLADD (A,B, CIN : in BIT; C : out BIT_VECTOR (1 downto 0) );
```

เมื่อเรียกใช้ ด้วย FULLADD(X1,X2,X3,Y);

ค่าของ X1 -> A X2 -> B X3 -> CIN Y <- C

- สัมพันธ์โดยชื่อ (Named Association) เป็นการบอกความสัมพันธ์ระหว่างชื่อกับชื่อ ด้วยการกำหนดว่าพารามิเตอร์ตัวส่งชื่ออะไร ต้องการส่งให้พารามิเตอร์ตัวรับชื่อว่าอะไร โดยใช้เครื่องหมาย " => " เช่น

```
FULLADD(A => X1, B => X2, CIN => X3, C => Y);
```

ตัวอย่างโปรซีเยอร์สำหรับวงจร Full adder เมื่อเขียนอยู่ใน Package

```
library ieee;
use ieee.std_logic_1164.all;
package my_package is
  procedure fulladd(A,B, CIN : in std_logic;
                    C : out std_logic_vector(1 downto 0) );
end my_package;
package body my_package is
  procedure fulladd (A,B, CIN : in std_logic;
                     C : out std_logic_VECTOR (1 downto 0)) is
  variable S, COUT : std_logic;
  begin
    S := A xor B xor CIN;
    COUT := (A and B) or (A and CIN) or (B and CIN);
    C := COUT & S;
  end fulladd;
end my_package;
```

การเรียกโปรซีเยอร์ FULLADD ใน package

```
library ieee;
use ieee.std_logic_1164.all;
use work.my_package.all;
entity EXAMPLE is
        port (  A,B : in std_logic_vector (3 downto 0);
                CIN : in std_logic;
                SUM : out std_logic_vector (3 downto 0);
                COUT : out std_logic );
end EXAMPLE;
architecture ARCHI of EXAMPLE is
begin
        process (A,B,CIN)
        variable S0, S1, S2, S3 : std_logic_vector (1 downto 0);
        begin
                fulladd (A(0), B(0), CIN, S0);
                fulladd (A(1), B(1), S0(1), S1);
                fulladd (A(2), B(2), S1(1), S2);
                fulladd (A(3), B(3), S2(1), S3);
                SUM <= S3(0) & S2(0) & S1(0) & S0(0);
                COUT <= S3(1);
        end process;
end ARCHI;
```

การแปลงประเภทของข้อมูล (Type conversion)

- ภาษา VHDL ค่อนข้างเข้มงวดเกี่ยวกับประเภทของข้อมูล กล่าวคือการส่งข้อมูลระหว่างออบเจ็กต์นั้น ออบเจ็กต์ต้องเป็นประเภทเดียวกัน เช่น BIT กับ BIT หรือ BIT_VECTOR กับ BIT_VECTOR จะเป็นคนละชนิดไม่ได้ เช่น BIT กับ STD_LOGIC หรือ STD_LOGIC_VECTOR กับ BIT_VECTOR ก็ไม่ได้ ดังนั้นใน LIBRARY IEEE จึงมี Package ที่ได้บรรจุฟังก์ชั่นเพื่อการแปลงข้อมูลบางประเภทไว้

ตัวอย่างฟังก์ชั่น

- ตัวอย่างฟังก์ชั่นที่อยู่ใน **std_logic_1164**

ฟังก์ชั่น	แปลงจาก	เป็น
TO_BIT	STD_ULOGIC	BIT
TO_BITVECTOR	STD_LOGIC_VECTOR	BIT_VECTOR

- ตัวอย่างฟังก์ชั่นที่อยู่ใน **Std_Logic_Arith**

ฟังก์ชั่น	แปลงจาก	เป็น
CONV_INTEGER	STD_ULOGIC	SMALL_INT
CONV_INTEGER	UNSIGNED	INTEGER

ตัวอย่างการใช้งาน

```
LIBRARY ieee;
USE ieee.std_logic_1164.ALL;
USE ieee.std_logic_arith.ALL;
USE ieee.std_logic_unsigned.ALL;
ENTITY ex_conversion IS
        PORT ( a, b : IN std_logic_vector(3 downto 0);
        c : OUT std_logic_vector(7 downto 0));
END ex_conversion;
ARCHITECTURE behav OF ex_conversion IS
SIGNAL a_i, b_i, c_i : integer range 0 to 255;
BEGIN
        -- convert from std_logic_vector to integer
        a_i <= (conv_integer(a));
        b_i <= (conv_integer(b));
        c_i <= a_i * b_i;
        -- convert from integer to a 8 bit std_logic_vector
        c <= (conv_std_logic_vector(c_i,8));
END behav;
```